Rules for integrands of the form $(a + b x^n)^p (c + d x^n)^q (e + f x^n)^r$

when $b c - a d \neq 0 \land b e - a f \neq 0 \land d e - c f \neq 0$

0: $\int (a + b x^n)^p (c + d x^n)^q (e + f x^n)^r \, dx$ when $(p \mid q \mid r) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.5.1: If $(p \mid q \mid r) \in \mathbb{Z}^+$, then

$$\int (a + b x^n)^p (c + d x^n)^q (e + f x^n)^r \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(a + b x^n)^p (c + d x^n)^q (e + f x^n)^r, x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[p,0] && IGtQ[q,0] && IGtQ[r,0]
```

1. $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)\,dx$

1: $\int \frac{e+f x^n}{(a+b x^n)(c+d x^n)}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{e+f z}{(a+b z)(c+d z)} == \frac{b e-a f}{(b c-a d)(a+b z)} - \frac{d e-c f}{(b c-a d)(c+d z)}$

Rule 1.1.3.5.1.1:

$$\int \frac{e+f x^n}{(a+b x^n)(c+d x^n)}\,dx \longrightarrow \frac{b e-a f}{b c-a d}\int \frac{1}{a+b x^n}\,dx - \frac{d e-c f}{b c-a d}\int \frac{1}{c+d x^n}\,dx$$

Program code:

```
Int[(e_+f_.*x_^n_)/((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[1/(a+b*x^n),x] -
  (d*e-c*f)/(b*c-a*d)*Int[1/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,e,f,n},x]
```

2: $\displaystyle\int \frac{e+f x^n}{(a+b x^n)\sqrt{c+d x^n}}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{e+f z}{a+b z} == \frac{f}{b} + \frac{b e-a f}{b\,(a+b z)}$

Rule 1.1.3.5.1.2:

$$\int \frac{e+f x^n}{(a+b x^n)\sqrt{c+d x^n}}\,dx \longrightarrow \frac{f}{b}\int \frac{1}{\sqrt{c+d x^n}}\,dx + \frac{b e-a f}{b}\int \frac{1}{(a+b x^n)\sqrt{c+d x^n}}\,dx$$

Program code:

```
Int[(e_+f_.*x_^n_)/((a_+b_.*x_^n_)*Sqrt[c_+d_.*x_^n_]),x_Symbol] :=
  f/b*Int[1/Sqrt[c+d*x^n],x] +
  (b*e-a*f)/b*Int[1/((a+b*x^n)*Sqrt[c+d*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,n},x]
```

3: $\int \frac{e+f x^n}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx$

Derivation: Algebraic expansion

Basis: $\frac{e+f z}{\sqrt{a+b z}} == \frac{f \sqrt{a+b z}}{b} + \frac{b e-a f}{b \sqrt{a+b z}}$

Rule 1.1.3.5.1.3:

$$\int \frac{e+f x^n}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx \longrightarrow \frac{f}{b} \int \frac{\sqrt{a+b x^n}}{\sqrt{c+d x^n}}\, dx + \frac{b e-a f}{b} \int \frac{1}{\sqrt{a+b x^n}\ \sqrt{c+d x^n}}\, dx$$

Program code:

```
Int[(e_+f_.*x_^n_)/(Sqrt[a_+b_.*x_^n_]*Sqrt[c_+d_.*x_^n_]),x_Symbol] :=
  f/b*Int[Sqrt[a+b*x^n]/Sqrt[c+d*x^n],x] +
  (b*e-a*f)/b*Int[1/(Sqrt[a+b*x^n]*Sqrt[c+d*x^n]),x] /;
FreeQ[{a,b,c,d,e,f,n},x] &&
  Not[EqQ[n,2] && (PosQ[b/a] && PosQ[d/c] || NegQ[b/a] && (PosQ[d/c] || GtQ[a,0] && (Not[GtQ[c,0]] || SimplerSqrtQ[-b/a,-d/c])))]
```

4. $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)\, dx$ when $p < -1$

1: $\int \frac{e+f x^2}{\sqrt{a+b x^2}\ (c+d x^2)^{3/2}}\, dx$ when $\frac{b}{a} > 0 \land \frac{d}{c} > 0$

Derivation: Algebraic expansion

Basis: $\frac{e+f x^2}{\sqrt{a+b x^2}\ (c+d x^2)^{3/2}} == \frac{b e-a f}{(b c-a d) \sqrt{a+b x^2}\ \sqrt{c+d x^2}} - \frac{(d e-c f) \sqrt{a+b x^2}}{(b c-a d) (c+d x^2)^{3/2}}$

Rule 1.1.3.5.1.4.1: If $\frac{b}{a} > 0 \land \frac{d}{c} > 0$, then

$$\int \frac{e+f x^2}{\sqrt{a+b x^2}\,\left(c+d x^2\right)^{3/2}}\,dx \longrightarrow \frac{b e-a f}{b c-a d}\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx-\frac{d e-c f}{b c-a d}\int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2}}\,dx$$

Program code:

```
Int[(e_+f_.*x_^2)/(Sqrt[a_+b_.*x_^2]*(c_+d_.*x_^2)^(3/2)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x] -
  (d*e-c*f)/(b*c-a*d)*Int[Sqrt[a+b*x^2]/(c+d*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[b/a] && PosQ[d/c]
```

2: $\int (a+b x^n)^p\,(c+d x^n)^q\,(e+f x^n)\,dx$ when $p<-1\land q>0$

Derivation: Binomial product recurrence 1 with $p = 0$

Rule 1.1.3.5.1.4.2: If $p<-1\land q>0$, then

$$\int (a+b x^n)^p\,(c+d x^n)^q\,(e+f x^n)\,dx \longrightarrow$$
$$-\frac{(b e-a f)\,x\,(a+b x^n)^{p+1}\,(c+d x^n)^q}{a b n\,(p+1)}+$$
$$\frac{1}{a b n\,(p+1)}\int (a+b x^n)^{p+1}\,(c+d x^n)^{q-1}\,\left(c\,(b e n\,(p+1)+b e-a f)+d\,(b e n\,(p+1)+(b e-a f)\,(n q+1))\,x^n\right)dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*x*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*b*n*(p+1)) +
  1/(a*b*n*(p+1))*
    Int[(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*Simp[c*(b*e*n*(p+1)+b*e-a*f)+d*(b*e*n*(p+1)+(b*e-a*f)*(n*q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && LtQ[p,-1] && GtQ[q,0]
```

3: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)\,dx$ when $p<-1$

Derivation: Binomial product recurrence 2a with $p=0$

Rule 1.1.3.5.1.4.3: If $p<-1$, then

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)\,dx \longrightarrow$$
$$-\frac{(b e-a f)\,x\,(a+b x^n)^{p+1}(c+d x^n)^{q+1}}{a n (b c-a d)(p+1)}+$$
$$\frac{1}{a n (b c-a d)(p+1)}\int (a+b x^n)^{p+1}(c+d x^n)^q\left(c (b e-a f)+e n (b c-a d)(p+1)+d (b e-a f)(n (p+q+2)+1) x^n\right)dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  -(b*e-a*f)*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*n*(b*c-a*d)*(p+1)) +
  1/(a*n*(b*c-a*d)*(p+1))*
    Int[(a+b*x^n)^(p+1)*(c+d*x^n)^q*Simp[c*(b*e-a*f)+e*n*(b*c-a*d)*(p+1)+d*(b*e-a*f)*(n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,n,q},x] && LtQ[p,-1]
```

5: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n) \, dx$ when $q>0 \land n(p+q+1)+1 \neq 0$

Derivation: Binomial product recurrence 3a with $p = 0$

Rule 1.1.3.5.1.5: If $q>0 \land n(p+q+1)+1 \neq 0$, then

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n) \, dx \to \frac{f x (a+b x^n)^{p+1} (c+d x^n)^q}{b (n (p+q+1)+1)} + \frac{1}{b (n (p+q+1)+1)} \int (a+b x^n)^p (c+d x^n)^{q-1} \left(c \left(b e - a f + b e n (p+q+1)\right) + \left(d \left(b e - a f\right) + f n q (b c - a d) + b d e n (p+q+1)\right) x^n\right) dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  f*x*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(b*(n*(p+q+1)+1)) +
  1/(b*(n*(p+q+1)+1))*
    Int[(a+b*x^n)^p*(c+d*x^n)^(q-1)*Simp[c*(b*e-a*f+b*e*n*(p+q+1))+(d*(b*e-a*f)+f*n*q*(b*c-a*d)+b*d*e*n*(p+q+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && GtQ[q,0] && NeQ[n*(p+q+1)+1,0]
```

6. $\int \frac{(a+b x^n)^p (e+f x^n)}{c+d x^n} \, dx$

1: $\int \frac{e+f x^4}{(a+b x^4)^{3/4} (c+d x^4)} \, dx$

Derivation: Algebraic expansion

Basis: $\frac{e+f z}{(a+b z)^{3/4} (c+d z)} == \frac{b e-a f}{(b c-a d) (a+b z)^{3/4}} - \frac{(d e-c f) (a+b z)^{1/4}}{(b c-a d) (c+d z)}$

Rule 1.1.3.5.1.6.1:

$$\int \frac{e+f x^4}{\left(a+b x^4\right)^{3/4} \left(c+d x^4\right)} \, dx \longrightarrow \frac{b e-a f}{b c-a d} \int \frac{1}{\left(a+b x^4\right)^{3/4}} \, dx - \frac{d e-c f}{b c-a d} \int \frac{\left(a+b x^4\right)^{1/4}}{c+d x^4} \, dx$$

–

Program code:

```
Int[(e_+f_.*x_^4)/((a_+b_.*x_^4)^(3/4)*(c_+d_.*x_^4)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[1/(a+b*x^4)^(3/4),x] - (d*e-c*f)/(b*c-a*d)*Int[(a+b*x^4)^(1/4)/(c+d*x^4),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

2: $\int \frac{\left(a+b x^n\right)^p \left(e+f x^n\right)}{c+d x^n} \, dx$

Derivation: Algebraic expansion

Basis: $\frac{e+f z}{c+d z} == \frac{f}{d} + \frac{d e-c f}{d (c+d z)}$

Rule 1.1.3.5.1.6.2:

$$\int \frac{\left(a+b x^n\right)^p \left(e+f x^n\right)}{c+d x^n} \, dx \longrightarrow \frac{f}{d} \int \left(a+b x^n\right)^p \, dx + \frac{d e-c f}{d} \int \frac{\left(a+b x^n\right)^p}{c+d x^n} \, dx$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(e_+f_.*x_^n_)/(c_+d_.*x_^n_),x_Symbol] :=
  f/d*Int[(a+b*x^n)^p,x] + (d*e-c*f)/d*Int[(a+b*x^n)^p/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,e,f,p,n},x]
```

7: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n) \, dx$

Derivation: Algebraic expansion

Rule 1.1.3.5.1.7:

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n) \, dx \longrightarrow e \int (a+b x^n)^p (c+d x^n)^q \, dx + f \int x^n (a+b x^n)^p (c+d x^n)^q \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_),x_Symbol] :=
  e*Int[(a+b*x^n)^p*(c+d*x^n)^q,x] + f*Int[x^n*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,e,f,n,p,q},x]
```

2. $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $p \in \mathbb{Z}^-$

1. $\int \frac{(c+d x^2)^q (e+f x^2)^r}{a+b x^2} \, dx$

1: $\int \frac{1}{(a+b x^2) (c+d x^2) \sqrt{e+f x^2}} \, dx$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b z) (c+d z)} == \frac{b}{(b c-a d) (a+b z)} - \frac{d}{(b c-a d) (c+d z)}$

Rule 1.1.3.5.2.1.1:

$$\int \frac{1}{(a+b x^2) (c+d x^2) \sqrt{e+f x^2}} \, dx \to \frac{b}{b c-a d} \int \frac{1}{(a+b x^2) \sqrt{e+f x^2}} \, dx - \frac{d}{b c-a d} \int \frac{1}{(c+d x^2) \sqrt{e+f x^2}} \, dx$$

Program code:

```
Int[1/((a_+b_.*x_^2)*(c_+d_.*x_^2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  b/(b*c-a*d)*Int[1/((a+b*x^2)*Sqrt[e+f*x^2]),x] -
  d/(b*c-a*d)*Int[1/((c+d*x^2)*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

```
Int[1/(x_^2*(c_+d_.*x_^2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  1/c*Int[1/(x^2*Sqrt[e+f*x^2]),x] -
  d/c*Int[1/((c+d*x^2)*Sqrt[e+f*x^2]),x] /;
FreeQ[{c,d,e,f},x] && NeQ[d*e-c*f,0]
```

2: $\displaystyle\int \frac{\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}{a+b\,x^2}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d\,z}}{a+b\,z} == \frac{d}{b\,\sqrt{c+d\,z}} + \frac{b\,c-a\,d}{b\,(a+b\,z)\,\sqrt{c+d\,z}}$

Rule 1.1.3.5.2.1.2:

$$\int \frac{\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}{a+b\,x^2}\,dx \longrightarrow \frac{d}{b}\int \frac{\sqrt{e+f\,x^2}}{\sqrt{c+d\,x^2}}\,dx + \frac{b\,c-a\,d}{b}\int \frac{\sqrt{e+f\,x^2}}{\left(a+b\,x^2\right)\sqrt{c+d\,x^2}}\,dx$$

Program code:

```
Int[Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]/(a_+b_.*x_^2),x_Symbol] :=
  d/b*Int[Sqrt[e+f*x^2]/Sqrt[c+d*x^2],x] + (b*c-a*d)/b*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[d/c,0] && GtQ[f/e,0] && Not[SimplerSqrtQ[d/c,f/e]]
```

```
Int[Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]/(a_+b_.*x_^2),x_Symbol] :=
  d/b*Int[Sqrt[e+f*x^2]/Sqrt[c+d*x^2],x] + (b*c-a*d)/b*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[SimplerSqrtQ[-f/e,-d/c]]
```

3. $\int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} dx$

1: $\int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} dx$ when $\frac{d}{c} > 0 \land \frac{f}{e} > 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b x^2) \sqrt{e+f x^2}} = -\frac{f}{(b e-a f) \sqrt{e+f x^2}} + \frac{b \sqrt{e+f x^2}}{(b e-a f) (a+b x^2)}$

Rule 1.1.3.5.2.1.3.1: If $\frac{d}{c} > 0 \land \frac{f}{e} > 0$, then

$$\int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} dx \to -\frac{f}{b e-a f} \int \frac{1}{\sqrt{c+d x^2} \sqrt{e+f x^2}} dx + \frac{b}{b e-a f} \int \frac{\sqrt{e+f x^2}}{(a+b x^2) \sqrt{c+d x^2}} dx$$

Program code:

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  -f/(b*e-a*f)*Int[1/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] +
  b/(b*e-a*f)*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[d/c,0] && GtQ[f/e,0] && Not[SimplerSqrtQ[d/c,f/e]]
```

2. $\int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} dx$ when $\frac{d}{c} \ngtr 0$

1: $\int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} dx$ when $\frac{d}{c} \ngtr 0 \land c > 0 \land e > 0$

Rule 1.1.3.5.2.1.3.2.1: If $\frac{d}{c} \ngtr 0 \land c > 0 \land e > 0$, then

$$\int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx \rightarrow \frac{1}{a \sqrt{c} \sqrt{e} \sqrt{-\frac{d}{c}}} \text{EllipticPi}\left[\frac{b c}{a d}, \text{ArcSin}\left[\sqrt{-\frac{d}{c}} x\right], \frac{c f}{d e}\right]$$

Program code:

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  1/(a*Sqrt[c]*Sqrt[e]*Rt[-d/c,2])*EllipticPi[b*c/(a*d), ArcSin[Rt[-d/c,2]*x], c*f/(d*e)] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[d/c,0]] && GtQ[c,0] && GtQ[e,0] && Not[Not[GtQ[f/e,0]] && SimplerSqrtQ[-f/e,-d/c]]
```

2: $\int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx$ when $\frac{d}{c} \not> 0 \land c \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} == 0$

Rule 1.1.3.5.2.1.3.2.2: If $\frac{d}{c} \not> 0 \land c \not> 0$, then

$$\int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx \rightarrow \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} \int \frac{1}{\left(a+b x^2\right) \sqrt{1+\frac{d}{c} x^2} \sqrt{e+f x^2}} \, dx$$

Program code:

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  Sqrt[1+d/c*x^2]/Sqrt[c+d*x^2]*Int[1/((a+b*x^2)*Sqrt[1+d/c*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[GtQ[c,0]]
```

4. $\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx$

1: $\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx$ when $\frac{d}{c} > 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{c+d x^2} \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}}{\sqrt{e+f x^2}} == 0$

Rule 1.1.3.5.2.1.4.1: If $\frac{d}{c} > 0$, then

$$\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx \to \frac{c \sqrt{e+f x^2}}{e \sqrt{c+d x^2} \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}} \int \frac{1}{\left(a+b x^2\right) \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}} \, dx \to \frac{c \sqrt{e+f x^2}}{a e \sqrt{\frac{d}{c}} \sqrt{c+d x^2} \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}} \text{EllipticPi}\left[1-\frac{b c}{a d}, \text{ArcTan}\left[\sqrt{\frac{d}{c}} x\right], 1-\frac{c f}{d e}\right]$$

$$\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx \to \frac{\sqrt{c+d x^2} \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}}{\sqrt{e+f x^2}} \int \frac{1}{\left(a+b x^2\right) \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}} \, dx \to \frac{\sqrt{c+d x^2} \sqrt{\frac{c (e+f x^2)}{e (c+d x^2)}}}{a \sqrt{\frac{d}{c}} \sqrt{e+f x^2}} \text{EllipticPi}\left[1-\frac{b c}{a d}, \text{ArcTan}\left[\sqrt{\frac{d}{c}} x\right], 1-\frac{c f}{d e}\right]$$

Program code:

```
Int[Sqrt[c_+d_.*x_^2]/((a_+b_.*x_^2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  c*Sqrt[e+f*x^2]/(a*e*Rt[d/c,2]*Sqrt[c+d*x^2]*Sqrt[c*(e+f*x^2)/(e*(c+d*x^2))])*
    EllipticPi[1-b*c/(a*d),ArcTan[Rt[d/c,2]*x],1-c*f/(d*e)] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[d/c]
```

```
(* Int[Sqrt[c_+d_.*x_^2]/((a_+b_.*x_^2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  Sqrt[c+d*x^2]*Sqrt[c*(e+f*x^2)/(e*(c+d*x^2))]/(a*Rt[d/c,2]*Sqrt[e+f*x^2])*
    EllipticPi[1-b*c/(a*d),ArcTan[Rt[d/c,2]*x],1-c*f/(d*e)] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[d/c] *)
```

2: $\displaystyle\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx$ when $\frac{d}{c} \ngtr 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{c+d x^2}}{a+b x^2} == \frac{d}{b \sqrt{c+d x^2}} + \frac{b c-a d}{b \left(a+b x^2\right) \sqrt{c+d x^2}}$

Rule 1.1.3.5.2.1.4.2: If $\frac{d}{c} \ngtr 0$, then

$$\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right) \sqrt{e+f x^2}} \, dx \rightarrow \frac{d}{b} \int \frac{1}{\sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx + \frac{b c-a d}{b} \int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx$$

Program code:

```
Int[Sqrt[c_+d_.*x_^2]/((a_+b_.*x_^2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  d/b*Int[1/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] +
  (b*c-a*d)/b*Int[1/((a+b*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NegQ[d/c]
```

5. $\displaystyle\int \frac{\left(c+d x^2\right)^q \left(e+f x^2\right)^r}{a+b x^2} \, dx$ when $q>0$

1: $\displaystyle\int \frac{\sqrt{e+f x^2}}{\left(a+b x^2\right) \left(c+d x^2\right)^{3/2}} \, dx$ when $\frac{d}{c}>0 \land \frac{f}{e}>0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\left(a+b x^2\right) \left(c+d x^2\right)^{3/2}} == \frac{b}{(b c-a d) \left(a+b x^2\right) \sqrt{c+d x^2}} - \frac{d}{(b c-a d) \left(c+d x^2\right)^{3/2}}$

Rule 1.1.3.5.2.1.5.1: If $\frac{d}{c}>0 \land \frac{f}{e}>0$, then

$$\int \frac{\sqrt{e+f x^2}}{\left(a+b x^2\right) \left(c+d x^2\right)^{3/2}} \, dx \longrightarrow \frac{b}{b c-a d} \int \frac{\sqrt{e+f x^2}}{\left(a+b x^2\right) \sqrt{c+d x^2}} \, dx - \frac{d}{b c-a d} \int \frac{\sqrt{e+f x^2}}{\left(c+d x^2\right)^{3/2}} \, dx$$

- Program code:

```
Int[Sqrt[e_+f_.*x_^2]/((a_+b_.*x_^2)*(c_+d_.*x_^2)^(3/2)),x_Symbol] :=
  b/(b*c-a*d)*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] -
  d/(b*c-a*d)*Int[Sqrt[e+f*x^2]/(c+d*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[d/c] && PosQ[f/e]
```

2: $\int \frac{\left(e+f x^2\right)^{3/2}}{\left(a+b x^2\right) \left(c+d x^2\right)^{3/2}} \, dx$ when $\frac{d}{c} > 0 \land \frac{f}{e} > 0$

- Derivation: Algebraic expansion

- Basis: $\frac{e+f x^2}{\left(a+b x^2\right) \left(c+d x^2\right)} == \frac{b e-a f}{(b c-a d) \left(a+b x^2\right)} - \frac{d e-c f}{(b c-a d) \left(c+d x^2\right)}$

- Rule 1.1.3.5.2.1.5.2: If $\frac{d}{c} > 0 \land \frac{f}{e} > 0$, then

$$\int \frac{\left(e+f x^2\right)^{3/2}}{\left(a+b x^2\right) \left(c+d x^2\right)^{3/2}} \, dx \longrightarrow \frac{b e-a f}{b c-a d} \int \frac{\sqrt{e+f x^2}}{\left(a+b x^2\right) \sqrt{c+d x^2}} \, dx - \frac{d e-c f}{b c-a d} \int \frac{\sqrt{e+f x^2}}{\left(c+d x^2\right)^{3/2}} \, dx$$

- Program code:

```
Int[(e_+f_.*x_^2)^(3/2)/((a_+b_.*x_^2)*(c_+d_.*x_^2)^(3/2)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] -
  (d*e-c*f)/(b*c-a*d)*Int[Sqrt[e+f*x^2]/(c+d*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[d/c] && PosQ[f/e]
```

3: $\int \frac{\left(c+d\,x^2\right)^{3/2}\sqrt{e+f\,x^2}}{a+b\,x^2}\,dx$ when $\frac{d}{c}>0\,\land\,\frac{f}{e}>0$

Derivation: Algebraic expansion

Basis: $\frac{\left(c+d\,x^2\right)^{3/2}}{a+b\,x^2} == \frac{(b\,c-a\,d)^2}{b^2\left(a+b\,x^2\right)\sqrt{c+d\,x^2}} + \frac{d\left(2\,b\,c-a\,d+b\,d\,x^2\right)}{b^2\sqrt{c+d\,x^2}}$

Rule 1.1.3.5.2.1.5.3: If $\frac{d}{c}>0\,\land\,\frac{f}{e}>0$, then

$$\int \frac{\left(c+d\,x^2\right)^{3/2}\sqrt{e+f\,x^2}}{a+b\,x^2}\,dx \longrightarrow \frac{(b\,c-a\,d)^2}{b^2}\int \frac{\sqrt{e+f\,x^2}}{\left(a+b\,x^2\right)\sqrt{c+d\,x^2}}\,dx + \frac{d}{b^2}\int \frac{\left(2\,b\,c-a\,d+b\,d\,x^2\right)\sqrt{e+f\,x^2}}{\sqrt{c+d\,x^2}}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^(3/2)*Sqrt[e_+f_.*x_^2]/(a_+b_.*x_^2),x_Symbol] :=
  (b*c-a*d)^2/b^2*Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x] +
  d/b^2*Int[(2*b*c-a*d+b*d*x^2)*Sqrt[e+f*x^2]/Sqrt[c+d*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[d/c] && PosQ[f/e]
```

4: $\int \frac{\left(c+d\,x^2\right)^q\left(e+f\,x^2\right)^r}{a+b\,x^2}\,dx$ when $q<-1\,\land\,r>1$

Derivation: Algebraic expansion

Basis: $\frac{\left(c+d\,x^2\right)^q\left(e+f\,x^2\right)}{a+b\,x^2} == \frac{b\,(b\,e-a\,f)\left(c+d\,x^2\right)^{q+2}}{(b\,c-a\,d)^2\left(a+b\,x^2\right)} - \frac{\left(c+d\,x^2\right)^q\left(2\,b\,c\,d\,e-a\,d^2\,e-b\,c^2\,f+d^2\,(b\,e-a\,f)\,x^2\right)}{(b\,c-a\,d)^2}$

Rule 1.1.3.5.2.1.5.4: If $q<-1\,\land\,r>1$, then

$$\int \frac{\left(c+d\,x^2\right)^q\left(e+f\,x^2\right)^r}{a+b\,x^2}\,dx \longrightarrow$$

$$\frac{b\,(b\,e-a\,f)}{(b\,c-a\,d)^2}\int\frac{(c+d\,x^2)^{q+2}\,(e+f\,x^2)^{r-1}}{a+b\,x^2}\,dx-\frac{1}{(b\,c-a\,d)^2}\int(c+d\,x^2)^q\,(e+f\,x^2)^{r-1}\,\left(2\,b\,c\,d\,e-a\,d^2\,e-b\,c^2\,f+d^2\,(b\,e-a\,f)\,x^2\right)\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^q_*(e_+f_.*x_^2)^r_/(a_+b_.*x_^2),x_Symbol] :=
  b*(b*e-a*f)/(b*c-a*d)^2*Int[(c+d*x^2)^(q+2)*(e+f*x^2)^(r-1)/(a+b*x^2),x] -
  1/(b*c-a*d)^2*Int[(c+d*x^2)^q*(e+f*x^2)^(r-1)*(2*b*c*d*e-a*d^2*e-b*c^2*f+d^2*(b*e-a*f)*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && LtQ[q,-1] && GtQ[r,1]
```

5: $\int\frac{(c+d\,x^2)^q\,(e+f\,x^2)^r}{a+b\,x^2}\,dx$ when $q>1$

Derivation: Algebraic expansion

Basis: $c+d\,z==\frac{d\,(a+b\,z)}{b}+\frac{b\,c-a\,d}{b}$

Rule 1.1.3.5.2.1.5.5: If $q>1$, then

$$\int\frac{(c+d\,x^2)^q\,(e+f\,x^2)^r}{a+b\,x^2}\,dx\longrightarrow\frac{d}{b}\int(c+d\,x^2)^{q-1}\,(e+f\,x^2)^r\,dx+\frac{b\,c-a\,d}{b}\int\frac{(c+d\,x^2)^{q-1}\,(e+f\,x^2)^r}{a+b\,x^2}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^q_*(e_+f_.*x_^2)^r_/(a_+b_.*x_^2),x_Symbol] :=
  d/b*Int[(c+d*x^2)^(q-1)*(e+f*x^2)^r,x] +
  (b*c-a*d)/b*Int[(c+d*x^2)^(q-1)*(e+f*x^2)^r/(a+b*x^2),x] /;
FreeQ[{a,b,c,d,e,f,r},x] && GtQ[q,1]
```

6. $\int\frac{(c+d\,x^2)^q\,(e+f\,x^2)^r}{a+b\,x^2}\,dx$ when $q\le-1$

1: $\int \frac{(c+d x^2)^q (e+f x^2)^r}{a+b x^2} \, dx$ when $q < -1$

Derivation: Algebraic expansion

- Basis: $\frac{(c+d x^2)^q}{a+b x^2} == \frac{b^2 (c+d x^2)^{q+2}}{(b c-a d)^2 (a+b x^2)} - \frac{d (2 b c-a d+b d x^2) (c+d x^2)^q}{(b c-a d)^2}$

- Rule 1.1.3.5.2.1.6.1: If $q < -1$, then

$$\int \frac{(c+d x^2)^q (e+f x^2)^r}{a+b x^2} \, dx \to \frac{b^2}{(b c-a d)^2} \int \frac{(c+d x^2)^{q+2} (e+f x^2)^r}{a+b x^2} \, dx - \frac{d}{(b c-a d)^2} \int (c+d x^2)^q (e+f x^2)^r (2 b c-a d+b d x^2) \, dx$$

- Program code:

```
Int[(c_+d_.*x_^2)^q_*(e_+f_.*x_^2)^r_/(a_+b_.*x_^2),x_Symbol] :=
  b^2/(b*c-a*d)^2*Int[(c+d*x^2)^(q+2)*(e+f*x^2)^r/(a+b*x^2),x] -
  d/(b*c-a*d)^2*Int[(c+d*x^2)^q*(e+f*x^2)^r*(2*b*c-a*d+b*d*x^2),x] /;
FreeQ[{a,b,c,d,e,f,r},x] && LtQ[q,-1]
```

2: $\int \frac{(c+d x^2)^q (e+f x^2)^r}{a+b x^2} \, dx$ when $q \leq -1$

Derivation: Algebraic expansion

Basis: $1 = -\frac{d (a+b z)}{b c-a d} + \frac{b (c+d z)}{b c-a d}$

Rule 1.1.3.5.2.1.6.2: If $q \leq -1$, then

$$\int \frac{(c+d x^2)^q (e+f x^2)^r}{a+b x^2} \, dx \longrightarrow -\frac{d}{b c-a d} \int (c+d x^2)^q (e+f x^2)^r \, dx + \frac{b}{b c-a d} \int \frac{(c+d x^2)^{q+1} (e+f x^2)^r}{a+b x^2} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^q_*(e_+f_.*x_^2)^r_/(a_+b_.*x_^2),x_Symbol] :=
  -d/(b*c-a*d)*Int[(c+d*x^2)^q*(e+f*x^2)^r,x] +
  b/(b*c-a*d)*Int[(c+d*x^2)^(q+1)*(e+f*x^2)^r/(a+b*x^2),x] /;
FreeQ[{a,b,c,d,e,f,r},x] && LeQ[q,-1]
```

2. $\int \frac{(c+d x^2)^p (e+f x^2)^p}{(a+b x^2)^2} \, dx$ when $-1 \le q < 0 \land -1 \le r < 0$

1: $\int \frac{\sqrt{c+d x^2} \sqrt{e+f x^2}}{(a+b x^2)^2} \, dx$

Rule 1.1.3.5.2.2.1:

$$\int \frac{\sqrt{c+d x^2} \sqrt{e+f x^2}}{(a+b x^2)^2} \, dx \longrightarrow$$

$$\frac{x \sqrt{c+d x^2} \sqrt{e+f x^2}}{2 a (a+b x^2)} + \frac{d f}{2 a b^2} \int \frac{a-b x^2}{\sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx + \frac{b^2 c e - a^2 d f}{2 a b^2} \int \frac{1}{(a+b x^2) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx$$

Program code:

```
Int[Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]/(a_+b_.*x_^2)^2,x_Symbol] :=
  x*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]/(2*a*(a+b*x^2)) +
  d*f/(2*a*b^2)*Int[(a-b*x^2)/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] +
  (b^2*c*e-a^2*d*f)/(2*a*b^2)*Int[1/((a+b*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

2: $\displaystyle\int \frac{1}{\left(a+b x^2\right)^2 \sqrt{c+d x^2}\, \sqrt{e+f x^2}}\, dx$

Rule 1.1.3.5.2.2.2:

$$\int \frac{1}{\left(a+b x^2\right)^2 \sqrt{c+d x^2}\, \sqrt{e+f x^2}}\, dx \longrightarrow$$

$$\frac{b^2 x \sqrt{c+d x^2}\, \sqrt{e+f x^2}}{2 a (b c-a d) (b e-a f) \left(a+b x^2\right)}-\frac{d f}{2 a (b c-a d) (b e-a f)} \int \frac{a+b x^2}{\sqrt{c+d x^2}\, \sqrt{e+f x^2}}\, dx+\frac{b^2 c e+3 a^2 d f-2 a b (d e+c f)}{2 a (b c-a d) (b e-a f)} \int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2}\, \sqrt{e+f x^2}}\, dx$$

Program code:

```
Int[1/((a_+b_.*x_^2)^2*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  b^2*x*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]/(2*a*(b*c-a*d)*(b*e-a*f)*(a+b*x^2)) -
  d*f/(2*a*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x^2)/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] +
  (b^2*c*e+3*a^2*d*f-2*a*b*(d*e+c*f))/(2*a*(b*c-a*d)*(b*e-a*f))*Int[1/((a+b*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

3: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $p \in \mathbb{Z}^- \land q > 0$

Derivation: Algebraic expansion

Basis: $c + d z == \frac{d (a+b z)}{b} + \frac{b c-a d}{b}$

Rule 1.1.3.5.2.4: If $p \in \mathbb{Z}^- \land q > 0$, then

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \longrightarrow$$
$$\frac{d}{b} \int (a+b x^n)^{p+1} (c+d x^n)^{q-1} (e+f x^n)^r \, dx + \frac{b c-a d}{b} \int (a+b x^n)^p (c+d x^n)^{q-1} (e+f x^n)^r \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_)^r_,x_Symbol] :=
  d/b*Int[(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*(e+f*x^n)^r,x] +
  (b*c-a*d)/b*Int[(a+b*x^n)^p*(c+d*x^n)^(q-1)*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,n,r},x] && ILtQ[p,0] && GtQ[q,0]
```

4: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $p\in\mathbb{Z}^-\wedge q\le -1$

Derivation: Algebraic expansion

Basis: $1 = -\frac{d\,(a+b\,z)}{b\,c-a\,d}+\frac{b\,(c+d\,z)}{b\,c-a\,d}$

Rule 1.1.3.5.2.5: If $p\in\mathbb{Z}^-\wedge q\le -1$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \longrightarrow$$
$$\frac{b}{b\,c-a\,d}\int (a+b\,x^n)^p\,(c+d\,x^n)^{q+1}\,(e+f\,x^n)^r\,dx-\frac{d}{b\,c-a\,d}\int (a+b\,x^n)^{p+1}\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_)^r_,x_Symbol] :=
  b/(b*c-a*d)*Int[(a+b*x^n)^p*(c+d*x^n)^(q+1)*(e+f*x^n)^r,x] -
  d/(b*c-a*d)*Int[(a+b*x^n)^(p+1)*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,n,q},x] && ILtQ[p,0] && LeQ[q,-1]
```

3. $\int \left(a+b x^2\right)^p \left(c+d x^2\right)^q \left(e+f x^2\right)^r dx$

1: $\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}\,\sqrt{e+f x^2}}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{c+d x^2}\,\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{\sqrt{e+f x^2}\,\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}} = 0$

Basis: $\frac{1}{\left(a+b x^2\right)^{3/2}\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}\,\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}} = \frac{1}{a}\,\text{Subst}\left[\frac{1}{\sqrt{1-\frac{(b c-a d)\,x^2}{c}}\,\sqrt{1-\frac{(b e-a f)\,x^2}{e}}},\,x,\,\frac{x}{\sqrt{a+b x^2}}\right]\partial_x \frac{x}{\sqrt{a+b x^2}}$

Rule 1.1.3.5.2.3.1:

$$\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}\,\sqrt{e+f x^2}}\,dx \to \frac{a\sqrt{c+d x^2}\,\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{c\sqrt{e+f x^2}\,\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}}\int \frac{1}{\left(a+b x^2\right)^{3/2}\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}\,\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}\,dx$$

$$\to \frac{\sqrt{c+d x^2}\,\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{c\sqrt{e+f x^2}\,\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1-\frac{(b c-a d)\,x^2}{c}}\,\sqrt{1-\frac{(b e-a f)\,x^2}{e}}}\,dx,\,x,\,\frac{x}{\sqrt{a+b x^2}}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  Sqrt[c+d*x^2]*Sqrt[a*(e+f*x^2)/(e*(a+b*x^2))]/(c*Sqrt[e+f*x^2]*Sqrt[a*(c+d*x^2)/(c*(a+b*x^2))])*
    Subst[Int[1/(Sqrt[1-(b*c-a*d)*x^2/c]*Sqrt[1-(b*e-a*f)*x^2/e]),x],x,x/Sqrt[a+b*x^2]] /;
FreeQ[{a,b,c,d,e,f},x]
```

2: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}\,\sqrt{e+f x^2}}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{c+d x^2}\,\sqrt{\frac{a\,(e+f x^2)}{e\,(a+b x^2)}}}{\sqrt{e+f x^2}\,\sqrt{\frac{a\,(c+d x^2)}{c\,(a+b x^2)}}} == 0$

- Basis: $\frac{1}{\sqrt{a+b x^2}\,\sqrt{\frac{a\,(c+d x^2)}{c\,(a+b x^2)}}\,\sqrt{\frac{a\,(e+f x^2)}{e\,(a+b x^2)}}} == \text{Subst}\left[\frac{1}{(1-b x^2)\,\sqrt{1-\frac{(b c-a d)\,x^2}{c}}\,\sqrt{1-\frac{(b e-a f)\,x^2}{e}}},\,x,\,\frac{x}{\sqrt{a+b x^2}}\right]\,\partial_x \frac{x}{\sqrt{a+b x^2}}$

Rule 1.1.3.5.2.3.2:

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}\,\sqrt{e+f x^2}}\,dx \to \frac{a\sqrt{c+d x^2}\,\sqrt{\frac{a\,(e+f x^2)}{e\,(a+b x^2)}}}{c\sqrt{e+f x^2}\,\sqrt{\frac{a\,(c+d x^2)}{c\,(a+b x^2)}}}\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{\frac{a\,(c+d x^2)}{c\,(a+b x^2)}}\,\sqrt{\frac{a\,(e+f x^2)}{e\,(a+b x^2)}}}\,dx$$

$$\to \frac{a\sqrt{c+d x^2}\,\sqrt{\frac{a\,(e+f x^2)}{e\,(a+b x^2)}}}{c\sqrt{e+f x^2}\,\sqrt{\frac{a\,(c+d x^2)}{c\,(a+b x^2)}}}\,\text{Subst}\left[\int \frac{1}{(1-b x^2)\,\sqrt{1-\frac{(b c-a d)\,x^2}{c}}\,\sqrt{1-\frac{(b e-a f)\,x^2}{e}}}\,dx,\,x,\,\frac{x}{\sqrt{a+b x^2}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/(Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  a*Sqrt[c+d*x^2]*Sqrt[a*(e+f*x^2)/(e*(a+b*x^2))]/(c*Sqrt[e+f*x^2]*Sqrt[a*(c+d*x^2)/(c*(a+b*x^2))])*
    Subst[Int[1/((1-b*x^2)*Sqrt[1-(b*c-a*d)*x^2/c]*Sqrt[1-(b*e-a*f)*x^2/e]),x],x,x/Sqrt[a+b*x^2]] /;
FreeQ[{a,b,c,d,e,f},x]
```

3: $\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right)^{3/2} \sqrt{e+f x^2}}\, dx$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{c+d x^2}\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{\sqrt{e+f x^2}\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}} == 0$

- Basis: $\frac{\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}}{\left(a+b x^2\right)^{3/2}\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}} == \frac{1}{a}\,\mathrm{Subst}\left[\frac{\sqrt{1-\frac{(b c-a d)\, x^2}{c}}}{\sqrt{1-\frac{(b e-a f)\, x^2}{e}}},\, x,\, \frac{x}{\sqrt{a+b x^2}}\right] \partial_x \frac{x}{\sqrt{a+b x^2}}$

Rule 1.1.3.5.2.3.3:

$$\int \frac{\sqrt{c+d x^2}}{\left(a+b x^2\right)^{3/2} \sqrt{e+f x^2}}\, dx \longrightarrow \frac{\sqrt{c+d x^2}\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{\sqrt{e+f x^2}\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}} \int \frac{\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}}{\left(a+b x^2\right)^{3/2}\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}\, dx \longrightarrow$$

$$\frac{\sqrt{c+d x^2}\sqrt{\frac{a\left(e+f x^2\right)}{e\left(a+b x^2\right)}}}{a\sqrt{e+f x^2}\sqrt{\frac{a\left(c+d x^2\right)}{c\left(a+b x^2\right)}}}\,\mathrm{Subst}\left[\int \frac{\sqrt{1-\frac{(b c-a d)\, x^2}{c}}}{\sqrt{1-\frac{(b e-a f)\, x^2}{e}}}\, dx,\, x,\, \frac{x}{\sqrt{a+b x^2}}\right]$$

Program code:

```
Int[Sqrt[c_+d_.*x_^2]/((a_+b_.*x_^2)^(3/2)*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  Sqrt[c+d*x^2]*Sqrt[a*(e+f*x^2)/(e*(a+b*x^2))]/(a*Sqrt[e+f*x^2]*Sqrt[a*(c+d*x^2)/(c*(a+b*x^2))])*
    Subst[Int[Sqrt[1-(b*c-a*d)*x^2/c]/Sqrt[1-(b*e-a*f)*x^2/e],x],x,x/Sqrt[a+b*x^2]] /;
FreeQ[{a,b,c,d,e,f},x]
```

4. $\int \frac{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}{\sqrt{e+f x^2}}\, dx$

1: $\int \frac{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}{\sqrt{e+f x^2}}\, dx$ when $\frac{d e-c f}{c} > 0$

Rule 1.1.3.5.2.3.4.1: If $\frac{d e-c f}{c} > 0$, then

$$\int \frac{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}{\sqrt{e+f x^2}}\, dx \to$$

$$\frac{d x \sqrt{a+b x^2}\ \sqrt{e+f x^2}}{2 f \sqrt{c+d x^2}} - \frac{c (d e-c f)}{2 f} \int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2} \sqrt{e+f x^2}}\, dx + \frac{b c (d e-c f)}{2 d f} \int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}\ \sqrt{e+f x^2}}\, dx - \frac{b d e-b c f-a d f}{2 d f} \int \frac{\sqrt{c+d x^2}}{\sqrt{a+b x^2}\ \sqrt{e+f x^2}}\, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]/Sqrt[e_+f_.*x_^2],x_Symbol] :=
  d*x*Sqrt[a+b*x^2]*Sqrt[e+f*x^2]/(2*f*Sqrt[c+d*x^2]) -
  c*(d*e-c*f)/(2*f)*Int[Sqrt[a+b*x^2]/((c+d*x^2)^(3/2)*Sqrt[e+f*x^2]),x] +
  b*c*(d*e-c*f)/(2*d*f)*Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] -
  (b*d*e-b*c*f-a*d*f)/(2*d*f)*Int[Sqrt[c+d*x^2]/(Sqrt[a+b*x^2]*Sqrt[e+f*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && PosQ[(d*e-c*f)/c]
```

2: $\int \frac{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}{\sqrt{e+f x^2}}\, dx$ when $\frac{d e-c f}{c} \ngtr 0$

Rule 1.1.3.5.2.3.4.2: If $\frac{d e-c f}{c} \ngtr 0$, then

$$\int \frac{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}{\sqrt{e+f x^2}}\, dx \to$$

$$\frac{x\sqrt{a+b\,x^2}\,\sqrt{c+d\,x^2}}{2\sqrt{e+f\,x^2}}+\frac{e\,(b\,e-a\,f)}{2\,f}\int\frac{\sqrt{c+d\,x^2}}{\sqrt{a+b\,x^2}\,\left(e+f\,x^2\right)^{3/2}}\,dx+$$
$$\frac{(b\,e-a\,f)\,(d\,e-2\,c\,f)}{2\,f^2}\int\frac{1}{\sqrt{a+b\,x^2}\,\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx-\frac{b\,d\,e-b\,c\,f-a\,d\,f}{2\,f^2}\int\frac{\sqrt{e+f\,x^2}}{\sqrt{a+b\,x^2}\,\sqrt{c+d\,x^2}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]/Sqrt[e_+f_.*x_^2],x_Symbol] :=
  x*Sqrt[a+b*x^2]*Sqrt[c+d*x^2]/(2*Sqrt[e+f*x^2]) +
  e*(b*e-a*f)/(2*f)*Int[Sqrt[c+d*x^2]/(Sqrt[a+b*x^2]*(e+f*x^2)^(3/2)),x] +
  (b*e-a*f)*(d*e-2*c*f)/(2*f^2)*Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x] -
  (b*d*e-b*c*f-a*d*f)/(2*f^2)*Int[Sqrt[e+f*x^2]/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NegQ[(d*e-c*f)/c]
```

5: $\int \frac{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}{\left(e+f x^2\right)^{3/2}}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b x^2}}{\left(e+f x^2\right)^{3/2}} = \frac{b}{f\sqrt{a+b x^2}\,\sqrt{e+f x^2}} - \frac{b e-a f}{f\sqrt{a+b x^2}\,\left(e+f x^2\right)^{3/2}}$

Rule 1.1.3.5.2.3.5:

$$\int \frac{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}{\left(e+f x^2\right)^{3/2}}\,dx \longrightarrow \frac{b}{f}\int \frac{\sqrt{c+d x^2}}{\sqrt{a+b x^2}\,\sqrt{e+f x^2}}\,dx - \frac{b e-a f}{f}\int \frac{\sqrt{c+d x^2}}{\sqrt{a+b x^2}\,\left(e+f x^2\right)^{3/2}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]/(e_+f_.*x_^2)^(3/2),x_Symbol] :=
  b/f*Int[Sqrt[c+d*x^2]/(Sqrt[a+b*x^2]*Sqrt[e+f*x^2]),x] -
  (b*e-a*f)/f*Int[Sqrt[c+d*x^2]/(Sqrt[a+b*x^2]*(e+f*x^2)^(3/2)),x] /;
FreeQ[{a,b,c,d,e,f},x]
```

4: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.5.3: If $n \in \mathbb{Z}^+$, let $u = \text{ExpandIntegrand}[(a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r,\,x]$, if u is a sum, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx \;\longrightarrow\; \int u\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_)^r_,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && IGtQ[n,0]
```

5: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $F[x] = -\text{Subst}\left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.1.3.5.4: If $n \in \mathbb{Z}^-$, then

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \rightarrow -\text{Subst}\left[\int \frac{(a+b x^{-n})^p (c+d x^{-n})^q (e+f x^{-n})^r}{x^2} \, dx, x, \frac{1}{x}\right]$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_*(e_+f_.*x_^n_)^r_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q*(e+f*x^(-n))^r/x^2,x],x,1/x] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && ILtQ[n,0]
```

U: $\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$

–

Rule 1.1.3.5.X:

$$\int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx \rightarrow \int (a+b x^n)^p (c+d x^n)^q (e+f x^n)^r \, dx$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e_+f_.*x_^n_)^r_.,x_Symbol] :=
  Unintegrable[(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r},x]
```

S: $\int (a+b\,u^n)^p\,(c+d\,u^n)^q\,(e+f\,u^n)^r\,dx$ when $u == g+h\,x$

Derivation: Integration by substitution

Rule 1.1.3.5.S: If $u == g+h\,x$, then

$$\int (a+b\,u^n)^p\,(c+d\,u^n)^q\,(e+f\,u^n)^r\,dx \longrightarrow \frac{1}{h}\,\text{Subst}\left[\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,(e+f\,x^n)^r\,dx,\,x,\,u\right]$$

Program code:

```
Int[(a_.+b_.*u_^n_)^p_.*(c_.+d_.*v_^n_)^q_.*(e_.+f_.*w_^n_)^r_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x^n)^p*(c+d*x^n)^q*(e+f*x^n)^r,x],x,u] /;
FreeQ[{a,b,c,d,e,f,p,n,q,r},x] && EqQ[u,v] && EqQ[u,w] && LinearQ[u,x] && NeQ[u,x]
```

6. $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$

1: $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$ when $q \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $q \in \mathbb{Z}$, then $(c+d\,x^{-n})^q == \frac{(d+c\,x^n)^q}{x^{n\,q}}$

Rule 1.1.3.5.5.1: If $q \in \mathbb{Z}$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx \longrightarrow \int \frac{(a+b\,x^n)^p\,(d+c\,x^n)^q\,(e+f\,x^n)^r}{x^{n\,q}}\,dx$$

Program code:

```
Int[(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  Int[(a+b*x^n)^p*(d+c*x^n)^q*(e+f*x^n)^r/x^(n*q),x] /;
FreeQ[{a,b,c,d,e,f,n,p,r},x] && EqQ[mn,-n] && IntegerQ[q]
```

2: $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$ when $p\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $p\in\mathbb{Z}$, then $(a+b\,x^n)^p == x^{n\,p}\,(b+a\,x^{-n})^p$

Rule 1.1.3.5.5.2: If $p\in\mathbb{Z}\,\land\,r\in\mathbb{Z}$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx\,\longrightarrow\,\int x^{n\,(p+r)}\,(b+a\,x^{-n})^p\,(c+d\,x^{-n})^q\,(f+e\,x^{-n})^r\,dx$$

Program code:

```
Int[(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  Int[x^(n*(p+r))*(b+a*x^(-n))^p*(c+d*x^(-n))^q*(f+e*x^(-n))^r,x] /;
FreeQ[{a,b,c,d,e,f,n,q},x] && EqQ[mn,-n] && IntegerQ[p] && IntegerQ[r]
```

3: $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx$ when $q \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\,(c+d\,x^{-n})^q}{(d+c\,x^n)^q} == 0$

Basis: $\frac{x^{n\,q}\,(c+d\,x^{-n})^q}{(d+c\,x^n)^q} == \frac{x^{n\,\text{FracPart}[q]}\,(c+d\,x^{-n})^{\text{FracPart}[q]}}{(d+c\,x^n)^{\text{FracPart}[q]}}$

Rule 1.1.3.5.5.3: If $q \notin \mathbb{Z}$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,(e+f\,x^n)^r\,dx \longrightarrow \frac{x^{n\,\text{FracPart}[q]}\,(c+d\,x^{-n})^{\text{FracPart}[q]}}{(d+c\,x^n)^{\text{FracPart}[q]}} \int \frac{(a+b\,x^n)^p\,(d+c\,x^n)^q\,(e+f\,x^n)^r}{x^{n\,q}}\,dx$$

Program code:

```
Int[(a_.+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_*(e_+f_.*x_^n_.)^r_.,x_Symbol] :=
  x^(n*FracPart[q])*(c+d*x^(-n))^FracPart[q]/(d+c*x^n)^FracPart[q]*Int[(a+b*x^n)^p*(d+c*x^n)^q*(e+f*x^n)^r/x^(n*q),x] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r},x] && EqQ[mn,-n] && Not[IntegerQ[q]]
```

Rules for integrands of the form $(a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r$

1. $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0$

1: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$

Derivation: Algebraic simplification

Basis: If $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$, then $\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r==\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r$

Rule: If $e_2\,f_1+e_1\,f_2==0\,\land\,(r\in\mathbb{Z}\,\lor\,e_1>0\,\land\,e_2>0)$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx\,\longrightarrow\,\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e1_+f1_.*x_^n2_.)^r_.*(e2_+f2_.*x_^n2_.)^r_.,x_Symbol] :=
  Int[(a+b*x^n)^p*(c+d*x^n)^q*(e1*e2+f1*f2*x^n)^r,x] /;
FreeQ[{a,b,c,d,e1,f1,e2,f2,n,p,q,r},x] && EqQ[n2,n/2] && EqQ[e2*f1+e1*f2,0] && (IntegerQ[r] || GtQ[e1,0] && GtQ[e2,0])
```

2: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx$ when $e_2\,f_1+e_1\,f_2==0$

Derivation: Piecewise constant extraction

Basis: If $e_2\,f_1+e_1\,f_2==0$, then $\partial_x\frac{\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r}{\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r}==0$

Rule: If $e_2\,f_1+e_1\,f_2==0$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,\left(e_1+f_1\,x^{n/2}\right)^r\,\left(e_2+f_2\,x^{n/2}\right)^r\,dx\,\longrightarrow$$

$$\frac{\left(e_1+f_1\,x^{n/2}\right)^{\text{FracPart}[r]}\left(e_2+f_2\,x^{n/2}\right)^{\text{FracPart}[r]}}{\left(e_1\,e_2+f_1\,f_2\,x^n\right)^{\text{FracPart}[r]}}\int\left(a+b\,x^n\right)^p\left(c+d\,x^n\right)^q\left(e_1\,e_2+f_1\,f_2\,x^n\right)^r\,dx$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.*(e1_+f1_.*x_^n2_.)^r_.*(e2_+f2_.*x_^n2_.)^r_.,x_Symbol] :=
  (e1+f1*x^(n/2))^FracPart[r]*(e2+f2*x^(n/2))^FracPart[r]/(e1*e2+f1*f2*x^n)^FracPart[r]*
    Int[(a+b*x^n)^p*(c+d*x^n)^q*(e1*e2+f1*f2*x^n)^r,x] /;
FreeQ[{a,b,c,d,e1,f1,e2,f2,n,p,q,r},x] && EqQ[n2,n/2] && EqQ[e2*f1+e1*f2,0]
```